JOB Add 3 improvements to my EA for Peter Gregory
ECN-compatible — such expert advisor have a special input parameter called ECN_Mode
1) Add trailing stop either standard one or one using ATR
2) Add Breakout filter logic to order entry as described below
3) Add third method of initiating a trade using the TDI Red Green indicator attached
Turtle logic does not buy/sell until the price is above/below a previous high/low 
Item 1 is trailing stop 
See examples below or if you have your own trailing stop let me know how it works
Item 2 Changing the order entry method from simple cross over of ROC2 over ROC 1 to add breakout logic below
Attached is a screen shot of what I am talking about breakout 2
It shows a green arrow when ROC2 crosses ROC1 this is a buy signal (signal candle ) and the buy order is generated on the completion of the signal candle (the next candle)
I want to change the signal from placing the buy/sell order automatically to being conditional on having conditions 1), 2),3) being met
In the example the nearest high was the candle before the signal candle or 1 candle before the signal candle 1) condition 1 x=1
Condition 2) The buy/sell signal will only be acted upon and order placed if the current price is 
(x) Pips above/below the nearest high/low stipulated in condition 1)
Condition 3) says that rather than order being placed immediately if conditions 1&2 are met that the order can be made if conditions 1&2 are met for the next (x) candles  that way if the current price dips for a few candles then proceeds above the nearest high/low the order is placed. if the current price does not go above/below the nearest high/low then the order is not placed
[bookmark: _GoBack]Inputs for item 2
1)  High or low (x) candles ago (in attached example x=1)
2)  Close of  signal candle is (x) pips above a high in a buy and below on a sell
3)  The number of candles (x) after the signal candle that condition 2) will initiate an order 
The closest logic I found on MQL5 is  https://www.mql5.com/en/code/10735 Turtle logic
Item 3 uses TDI Red Green indicator to initiate a trade
When Green crosses above red it opens a buy trade with the logic of item 2 trade closes by one of 4 methods depending on which is triggered first  SL , break even, trailing stop or TP
When green crosses below Red a sell trade is initiated using same logic in item 2 and same closing logic as buy trade	Comment by Peter Gregory: 
Files:









  



breakout.zip


Breakout Filter.bmp








image3.emf
Turtle.zip


Turtle.zip


TheClassicTurtleTrader.mq4

//+------------------------------------------------------------------+

//| TheClassicTurtleTrader.mq4

//| Copyright © Pointzero-indicator.com

//+------------------------------------------------------------------+

#property copyright "Copyright © Pointzero-indicator.com"

#property link      "http://www.pointzero-indicator.com"



//---- indicator settings

#property indicator_chart_window

#property indicator_buffers 4

#property indicator_color1 DodgerBlue

#property indicator_color2 Red

#property indicator_color3 DodgerBlue

#property indicator_color4 Red

#property indicator_width1 2

#property indicator_width2 2

#property indicator_width3 3

#property indicator_width4 3

#property indicator_style1 STYLE_DOT

#property indicator_style2 STYLE_DOT



//---- constants to know if we are in or out of the market

#define IN_THE_MARKET 0

#define OUT_OF_MARKET 1



//---- indicator parameters

extern int  TradePeriod         = 20;        // Donchian channel period for trading signals

extern int  StopPeriod          = 10;        // Donchian channel period for exit signals

extern bool StrictEntry         = false;     // Apply strict entry parameters like the Turtles did

extern bool StrictExit          = false;     // Apply strict exit parameters like the Turtles did

extern bool StrictStop          = false;     // Apply strict stop-loss like the turtled did

extern bool Greedy              = false;     // Do not exit a trade unless it is in profit or the SL is hit

extern bool EvaluateStoploss    = true;      // Check if we have been stopped out and show future signals

extern int  ATRPeriod           = 20;        // ATRPeriod to set the stop-loss

extern double ATRStopNumber     = 2;         // N.Factor to calculate the stop-loss

extern bool DisplayAlerts       = false;     // You know...



//---- indicator buffers

double ExtMapBuffer1[];

double ExtMapBuffer2[];

double ExtMapBuffer3[];

double ExtMapBuffer4[];

double LongInfo[];

double ShortInfo[];

double TrendDirection[];



//---- alerts

static datetime TimeStamp;

static int AlertCount = 1;



//---- internal

double long_price;

double short_price;

double long_stop;

double short_stop;



//+------------------------------------------------------------------+

//| Custom indicator initialization function                         |

//+------------------------------------------------------------------+

int init()

{

   // Three more buffers to store long/short trading info and trend

   IndicatorBuffers(7);

   

   // Drawing settings

   SetIndexStyle(0,DRAW_ARROW); SetIndexArrow(0, 233);

   SetIndexStyle(1,DRAW_ARROW); SetIndexArrow(1, 234);

   SetIndexStyle(2,DRAW_ARROW); SetIndexArrow(2,159);

   SetIndexStyle(3,DRAW_ARROW); SetIndexArrow(3,159);

   IndicatorDigits(MarketInfo(Symbol(),MODE_DIGITS));

   

   // Name and labels

   IndicatorShortName("The Classic Turtle Trader ("+ TradePeriod +"-"+ StopPeriod +")");

   SetIndexLabel(0,"Buy breakout "+ TradePeriod);

   SetIndexLabel(1,"Sell breakout "+ TradePeriod);

   SetIndexLabel(2,"Exit longs "+ StopPeriod);

   SetIndexLabel(3,"Exit shorts "+ StopPeriod);

   

   // Buffers

   SetIndexBuffer(0,ExtMapBuffer1);

   SetIndexBuffer(1,ExtMapBuffer2);

   SetIndexBuffer(2,ExtMapBuffer3);    

   SetIndexBuffer(3,ExtMapBuffer4); 

   SetIndexBuffer(4,LongInfo);

   SetIndexBuffer(5,ShortInfo);

   SetIndexBuffer(6,TrendDirection);

   

   Comment("Copyright © http://www.pointzero-indicator.com");

   return(0);

}

//+------------------------------------------------------------------+

//| Custom indicator iteration function                              |

//+------------------------------------------------------------------+

int start()

{

     // More vars here too...

     int start = 0;

     int limit;

     int counted_bars = IndicatorCounted();



     // check for possible errors

     if(counted_bars < 0) 

        return(-1);

        

     // Only check these

     limit = Bars - 1 - counted_bars;

     if(counted_bars==0) limit-=1+1;

     // Check the signal foreach bar

     for(int i = limit; i >= start; i--)

     {           

         // Highs and lows

         double rhigh = iHigh(Symbol(),Period(),iHighest(Symbol(), Period(), MODE_HIGH, TradePeriod,i+1));

         double rlow  = iLow(Symbol(),Period(),iLowest(Symbol(), Period(), MODE_LOW, TradePeriod, i+1));

         double shigh = iHigh(Symbol(),Period(),iHighest(Symbol(), Period(), MODE_HIGH, StopPeriod,i+1));

         double slow  = iLow(Symbol(),Period(),iLowest(Symbol(), Period(), MODE_LOW, StopPeriod, i+1));

         

         // Candle value

         double CLOSE = iClose(Symbol(),0, i);

         double HIGH = iHigh(Symbol(),0, i);

         double LOW = iLow(Symbol(),0, i);

         

         // Stop-losses if we traded here

         double ATRv = iATR(Symbol(), 0, ATRPeriod, i) * ATRStopNumber;

         double ATRup = CLOSE + ATRv;

         double ATRlow = CLOSE - ATRv;

         

         // Respect old trend

         LongInfo[i] = LongInfo[i+1];

         ShortInfo[i] = ShortInfo[i+1];

         TrendDirection[i] = TrendDirection[i+1];

         

         // It might be recalculating bar zero

         ExtMapBuffer1[i] = EMPTY_VALUE;

         ExtMapBuffer2[i] = EMPTY_VALUE;

         ExtMapBuffer3[i] = EMPTY_VALUE;

         ExtMapBuffer4[i] = EMPTY_VALUE;

         

         //

         // 1) Check for exits

         //

             

         // Exit longs taking into account strict exits and the greedy mode

         // Loose mode only applies after the bar has closed

         if(((CLOSE < slow && i > 0) && (CLOSE > long_price || Greedy == false)) || 

            (LOW < slow && StrictExit == true && (LOW > long_price || Greedy == false)) && 

            LongInfo[i] == IN_THE_MARKET)

         {

            LongInfo[i] = OUT_OF_MARKET;

            ExtMapBuffer3[i] = LOW;

         }

         

         // Exit shorts taking into account strict exits and the greedy mode

         // Loose mode only applies after the bar has closed

         if(((CLOSE > shigh && i > 0) && (CLOSE < short_price || Greedy == false)) || 

            (HIGH > shigh && StrictExit == true && (HIGH < short_price || Greedy == false)) &&  

            ShortInfo[i] == IN_THE_MARKET)

         {

            ShortInfo[i] = OUT_OF_MARKET;

            ExtMapBuffer4[i] = HIGH;

         }

         

         //

         // 2) Check for stoploss hits

         //

         

         // Exit longs taking into account strict stoploss

         // Loose mode only applies after the bar has closed

         if(((CLOSE < long_stop && i > 0) || (LOW < long_stop && StrictStop == true)) && LongInfo[i] == IN_THE_MARKET && EvaluateStoploss == true)

         {

            LongInfo[i] = OUT_OF_MARKET;

            ExtMapBuffer3[i] = LOW;

         }

         

         // Exit shorts taking into account strict stoploss

         // Loose mode only applies after the bar has closed

         if(((CLOSE > short_stop && i > 0) || (HIGH > short_stop && StrictStop == true)) &&  ShortInfo[i] == IN_THE_MARKET && EvaluateStoploss == true)

         {

            ShortInfo[i] = OUT_OF_MARKET;

            ExtMapBuffer4[i] = HIGH;

         }

         

         //

         // 3) Check for entries

         //

         

         // Enter long trade taking into account strict entries

         // Loose mode only applies after the bar has closed

         if(((CLOSE > rhigh && i > 0) || (HIGH > rhigh && StrictEntry == true)) &&  LongInfo[i] == OUT_OF_MARKET)

         {

            TrendDirection[i] = OP_BUY;

            LongInfo[i] = IN_THE_MARKET;

            ExtMapBuffer1[i] = LOW;

            long_stop = ATRlow;

            long_price = CLOSE;

         }

         

         // Enter short trade taking into account strict entries

         if(((CLOSE < rlow && i > 0) || (LOW < rlow && StrictEntry == true)) && ShortInfo[i] == OUT_OF_MARKET)

         {

            TrendDirection[i] = OP_SELL;

            ShortInfo[i] = IN_THE_MARKET;

            ExtMapBuffer2[i] = HIGH;

            short_stop = ATRup;

            short_price = CLOSE;

         }

     }

     

     // Alerts

     if(TimeStamp != Time[0] && DisplayAlerts == true)

     {

         // Exit signals

         if(LongInfo[1] == OUT_OF_MARKET && LongInfo[2] == IN_THE_MARKET && AlertCount == 0)

         {

            Alert("[Classic Turtle Trader "+ TradePeriod +"-"+ StopPeriod +"]["+ Symbol() +"] Close longs");

         } else if (ShortInfo[1] == OUT_OF_MARKET && ShortInfo[2] == IN_THE_MARKET && AlertCount == 0) {

            Alert("[Classic Turtle Trader "+ TradePeriod +"-"+ StopPeriod +"]["+ Symbol() +"] Close shorts");

         }

         

         // Enter signals

         if(LongInfo[1] == IN_THE_MARKET && LongInfo[2] == OUT_OF_MARKET && AlertCount == 0)

         {

            Alert("[Classic Turtle Trader "+ TradePeriod +"-"+ StopPeriod +"]["+ Symbol() +"] BUY");

         } else if (ShortInfo[1] == IN_THE_MARKET && ShortInfo[2] == OUT_OF_MARKET && AlertCount == 0) {

            Alert("[Classic Turtle Trader "+ TradePeriod +"-"+ StopPeriod +"]["+ Symbol() +"] SELL");

         }

         TimeStamp = Time[0];

         AlertCount = 0;

     }

    

   // Bye Bye

   return(0);

}

ATR-Trailer.zip


ATR-Trailer.mq4

//+------------------------------------------------------------------+

//|                                                  ATR-Trailer.mq4 |

//|                                                    Andriy Moraru |

//|                                         http://www.earnforex.com |

//|            							                       2011-2013 |

//+------------------------------------------------------------------+

#property copyright "www.EarnForex.com, 2011-2013"

#property link      "http://www.earnforex.com"



// Plain trailing stop EA with ATR-based stop-loss.



#define LONG 1

#define SHORT 2



extern int ATR_Period = 24;

extern int ATR_Multiplier = 3;

extern int StartWith = 1; // 1 - Short, 2 - Long

extern int Slippage = 100; 	// Tolerated slippage in pips

extern double Lots = 0.1;

extern bool ECN_Mode = false; // Set to true if stop-loss should be added only after Position Open

extern int TakeProfit = 0; // In your broker's pips



extern int Magic = 123123123;



// Global variables

bool HaveLongPosition;

bool HaveShortPosition;



int LastPosition = 0;



int init()

{

   LastPosition = 3 - StartWith;

   return(0);

}



//+------------------------------------------------------------------+

//| Expert Every Tick Function                                       |

//+------------------------------------------------------------------+

int start()

{

	double SL = 0, TP = 0;

	

	if (IsTradeAllowed() == false) return(0);

	

   // Getting the ATR values

   double ATR = iATR(NULL, 0, ATR_Period, 0);

   ATR *= ATR_Multiplier;



   if (ATR <= (MarketInfo(Symbol(), MODE_STOPLEVEL) + MarketInfo(Symbol(), MODE_SPREAD)) * Point) ATR = (MarketInfo(Symbol(), MODE_STOPLEVEL) + MarketInfo(Symbol(), MODE_SPREAD)) * Point;



  	// Check what position is currently open

 	GetPositionStates();

   

 	// Adjust SL and TP of the current position

 	if ((HaveLongPosition) || (HaveShortPosition)) AdjustSLTP(ATR);

   else

	{

   	// Buy condition

   	if (LastPosition == SHORT)

   	{

  			for (int i = 0; i < 10; i++)

  			{

  			   RefreshRates();

  				// Bid and Ask are swapped to preserve the probabilities and decrease/increase profit/loss size

  		   	if (!ECN_Mode)

  		   	{

               SL = NormalizeDouble(Bid - ATR, Digits);

               if (TakeProfit > 0) TP = NormalizeDouble(Bid + TakeProfit * Point, Digits);

  		   	}

  				int result = OrderSend(Symbol(), OP_BUY, Lots, Ask, Slippage, SL, TP, "ATR-Trader", Magic, 0, Blue); 

  				Sleep(1000);

            if (result == -1)

            {

               int e = GetLastError();

               Print(e);

            }

  				else return(0);

  			}

   	}

   	// Sell condition

   	else if (LastPosition == LONG)

   	{

  			for (i = 0; i < 10; i++)

  			{

  			   RefreshRates();

  				// Bid and Ask are swapped to preserve the probabilities and decrease/increase profit/loss size

  		      if (!ECN_Mode)

  		      {

               SL = NormalizeDouble(Ask + ATR, Digits);

               if (TakeProfit > 0) TP = NormalizeDouble(Ask - TakeProfit * Point, Digits);

            }

  				result = OrderSend(Symbol(), OP_SELL, Lots, Bid, Slippage, SL, TP, "ATR-Trader", Magic, 0, Red); 

  				Sleep(1000);

            if (result == -1)

            {

               e = GetLastError();

               Print(e);

            }

  				else return(0);

  			}

   	}

	}

	

	return(0);

}



//+------------------------------------------------------------------+

//| Check What Position is Currently Open										|

//+------------------------------------------------------------------+

void GetPositionStates()

{

   int total = OrdersTotal();

   for (int cnt = 0; cnt < total; cnt++)

   {

      if (OrderSelect(cnt, SELECT_BY_POS, MODE_TRADES) == false) continue;

      if (OrderMagicNumber() != Magic) continue;

      if (OrderSymbol() != Symbol()) continue;



      if (OrderType() == OP_BUY)

      {

			HaveLongPosition = true;

			HaveShortPosition = false;

		}

      else if (OrderType() == OP_SELL)

      {

			HaveLongPosition = false;

			HaveShortPosition = true;

		}

   	if (HaveLongPosition) LastPosition = LONG;

   	else if (HaveShortPosition) LastPosition = SHORT;

   	return;

	}



   HaveLongPosition = false;

	HaveShortPosition = false;

}



//+------------------------------------------------------------------+

//| Adjust Stop-Loss and TakeProfit of the Open Position					|

//+------------------------------------------------------------------+

void AdjustSLTP(double SLparam)

{

   double TP = 0;

   

   int total = OrdersTotal();

   for (int cnt = 0; cnt < total; cnt++)

   {

      if (OrderSelect(cnt, SELECT_BY_POS, MODE_TRADES) == false) continue;

      if (OrderMagicNumber() != Magic) continue;

      if (OrderSymbol() != Symbol()) continue;



      if (OrderType() == OP_BUY)

		{

		   RefreshRates();

			double SL = NormalizeDouble(Bid - SLparam, Digits);

			if (TakeProfit > 0) TP = NormalizeDouble(Bid + TakeProfit * Point, Digits);

			if ((SL > NormalizeDouble(OrderStopLoss(), Digits)) || (NormalizeDouble(OrderStopLoss(), Digits) == 0) || ((TP > 0) && (NormalizeDouble(OrderTakeProfit(), Digits) == 0)))

			{

				for (int i = 0; i < 10; i++)

				{

      		   bool result = OrderModify(OrderTicket(), OrderOpenPrice(), SL, TP, 0);

      		   if (result) return;

				}

			}

		}

      else if (OrderType() == OP_SELL)

		{ 

		   RefreshRates();

			SL = NormalizeDouble(Ask + SLparam, Digits);

			if (TakeProfit > 0) TP = NormalizeDouble(Ask - TakeProfit * Point, Digits);

			if ((SL < NormalizeDouble(OrderStopLoss(), Digits)) || (NormalizeDouble(OrderStopLoss(), Digits) == 0) || ((TP > 0) && (NormalizeDouble(OrderTakeProfit(), Digits) == 0)))

			{

				for (i = 0; i < 10; i++)

				{

      		   result = OrderModify(OrderTicket(), OrderOpenPrice(), SL, TP, 0);

      		   if (result) return;

				}

			}

		}

	}

}

//+------------------------------------------------------------------+

Trailing Stop.mq4

#property copyright "Copyright © 2009-2015, EarnForex.com"

#property link      "http://www.earnforex.com"



/*

   Kicks in when position reaches at least TrailingStop pips of profit.

*/

 

extern double TrailingStop = 5;

 

// Set it to some value above 0 to activate stop-loss

extern double StopLoss = 0; 

 

int init()

{

   return(0);

}

 

int deinit()

{

   return(0);

}

 

int start()

{

  double PointValue;

  for (int i = 0; i < OrdersTotal(); i++) 

  {

      OrderSelect(i, SELECT_BY_POS, MODE_TRADES);

      if (OrderSymbol() != Symbol()) continue; // Skipping positions in other currency pairs

      //Calculate the point value in case there are extra digits in the quotes

      if (MarketInfo(OrderSymbol(), MODE_POINT) == 0.00001) PointValue = 0.0001;

      else if (MarketInfo(OrderSymbol(), MODE_POINT) == 0.001) PointValue = 0.01;

      else PointValue = MarketInfo(OrderSymbol(), MODE_POINT);

      //Normalize trailing stop value to the point value

      double TSTP = TrailingStop * PointValue;

 

      if (OrderType() == OP_BUY)

      {

         if (Bid - OrderOpenPrice() > TSTP)

         {

            if (OrderStopLoss() < Bid - TSTP)

            {

               if (!OrderModify(OrderTicket(), OrderOpenPrice(), Bid - TSTP, OrderTakeProfit(), OrderExpiration(), clrNONE))

                  Print("Error setting Buy trailing stop: ", GetLastError());

            }

         }

         else if ((OrderStopLoss() != Bid - StopLoss * PointValue) && (StopLoss != 0) && (OrderStopLoss() == 0))

            if (!OrderModify(OrderTicket(), OrderOpenPrice(), Bid - StopLoss * PointValue, OrderTakeProfit(), OrderExpiration(), clrNONE))

               Print("Error setting Buy stop-loss: ", GetLastError());

      }

      else if (OrderType() == OP_SELL)

      {

         if (OrderOpenPrice() - Ask > TSTP)

         {

            if ((OrderStopLoss() > Ask + TSTP) || (OrderStopLoss() == 0))

            {

               if (!OrderModify(OrderTicket(), OrderOpenPrice(), Ask + TSTP, OrderTakeProfit(), OrderExpiration(), clrNONE))

                  Print("Error setting Sell trailing stop: ", GetLastError());

            }

         }

         else if ((OrderStopLoss() != Ask + StopLoss * PointValue) && (StopLoss != 0) && (OrderStopLoss() == 0))

            if (!OrderModify(OrderTicket(), OrderOpenPrice(), Ask + StopLoss * PointValue, OrderTakeProfit(), OrderExpiration(), clrNONE))

               Print("Error setting Sell stop-loss: ", GetLastError());

      }

	}

 

   return(0);

}





image7.emf
TDI Red Green.mq4


TDI Red Green.mq4
//+------------------------------------------------------------------+

//|                                    Traders Dynamic Index.mq4     |

//|                                    Copyright © 2006, Dean Malone |

//|                                    www.compassfx.com             |

//+------------------------------------------------------------------+



//+------------------------------------------------------------------+

//|                                                                  |

//|                     Traders Dynamic Index                        |

//|                                                                  |

//|  This hybrid indicator is developed to assist traders in their   |

//|  ability to decipher and monitor market conditions related to    |

//|  trend direction, market strength, and market volatility.        |

//|                                                                  | 

//|  Even though comprehensive, the T.D.I. is easy to read and use.  |

//|                                                                  |

//|  Green line = RSI Price line                                     |

//|  Red line = Trade Signal line                                    |

//|  Blue lines = Volatility Band                                    | 

//|  Yellow line = Market Base Line                                  |  

//|                                                                  |

//|  Trend Direction - Immediate and Overall                         |

//|   Immediate = Green over Red...price action is moving up.        |

//|               Red over Green...price action is moving down.      |

//|                                                                  |   

//|   Overall = Yellow line trends up and down generally between the |

//|             lines 32 & 68. Watch for Yellow line to bounces off  |

//|             these lines for market reversal. Trade long when     |

//|             price is above the Yellow line, and trade short when |

//|             price is below.                                      |        

//|                                                                  |

//|  Market Strength & Volatility - Immediate and Overall            |

//|   Immediate = Green Line - Strong = Steep slope up or down.      | 

//|                            Weak = Moderate to Flat slope.        |

//|                                                                  |               

//|   Overall = Blue Lines - When expanding, market is strong and    |

//|             trending. When constricting, market is weak and      |

//|             in a range. When the Blue lines are extremely tight  |                                                       

//|             in a narrow range, expect an economic announcement   | 

//|             or other market condition to spike the market.       |

//|                                                                  |               

//|                                                                  |

//|  Entry conditions                                                |

//|   Scalping  - Long = Green over Red, Short = Red over Green      |

//|   Active - Long = Green over Red & Yellow lines                  |

//|            Short = Red over Green & Yellow lines                 |    

//|   Moderate - Long = Green over Red, Yellow, & 50 lines           |

//|              Short= Red over Green, Green below Yellow & 50 line |

//|                                                                  |

//|  Exit conditions*                                                |   

//|   Long = Green crosses below Red                                 |

//|   Short = Green crosses above Red                                |

//|   * If Green crosses either Blue lines, consider exiting when    |

//|     when the Green line crosses back over the Blue line.         |

//|                                                                  |

//|                                                                  |

//|  IMPORTANT: The default settings are well tested and proven.     |

//|             But, you can change the settings to fit your         |

//|             trading style.                                       |

//|                                                                  |

//|                                                                  |

//|  Price & Line Type settings:                           |                

//|   RSI Price settings                                             |               

//|   0 = Close price     [DEFAULT]                                  |               

//|   1 = Open price.                                                |               

//|   2 = High price.                                                |               

//|   3 = Low price.                                                 |               

//|   4 = Median price, (high+low)/2.                                |               

//|   5 = Typical price, (high+low+close)/3.                         |               

//|   6 = Weighted close price, (high+low+close+close)/4.            |               

//|                                                                  |               

//|   RSI Price Line & Signal Line Type settings                                   |               

//|   0 = Simple moving average       [DEFAULT]                      |               

//|   1 = Exponential moving average                                 |               

//|   2 = Smoothed moving average                                    |               

//|   3 = Linear weighted moving average                             |               

//|                                                                  |

//|   Good trading,                                                  |   

//|                                                                  |

//|   Dean                                                           |                              

//+------------------------------------------------------------------+







#property indicator_buffers 6

#property indicator_color1 Black

#property indicator_color2 0XFFFFFFFF  // MediumBlue

#property indicator_color3 0xFFFFFFFF  // Yellow

#property indicator_color4 0XFFFFFFFF  // MediumBlue

#property indicator_color5 Green

#property indicator_color6 Red

#property indicator_separate_window



extern int RSI_Period = 13;         //8-25

extern int RSI_Price = 0;           //0-6

extern int Volatility_Band = 34;    //20-40

extern int RSI_Price_Line = 2;      

extern int RSI_Price_Type = 0;      //0-3

extern int Trade_Signal_Line = 7;   

extern int Trade_Signal_Type = 0;   //0-3



double RSIBuf[],UpZone[],MdZone[],DnZone[],MaBuf[],MbBuf[];



int init()

  {

   IndicatorShortName("TDI");

   SetIndexBuffer(0,RSIBuf);

   SetIndexBuffer(1,UpZone);

   SetIndexBuffer(2,MdZone);

   SetIndexBuffer(3,DnZone);

   SetIndexBuffer(4,MaBuf);

   SetIndexBuffer(5,MbBuf);

   

   SetIndexStyle(0,DRAW_NONE); 

   SetIndexStyle(1,DRAW_LINE); 

   SetIndexStyle(2,DRAW_LINE);   //,0,2

   SetIndexStyle(3,DRAW_LINE);

   SetIndexStyle(4,DRAW_LINE);   //,0,2

   SetIndexStyle(5,DRAW_LINE);   //,0,2

   

   SetIndexLabel(0,NULL); 

   SetIndexLabel(1,"VB High"); 

   SetIndexLabel(2,"Market Base Line"); 

   SetIndexLabel(3,"VB Low"); 

   SetIndexLabel(4,"RSI Price Line");

   SetIndexLabel(5,"Trade Signal Line");

 

   SetLevelValue(0,50);

   SetLevelValue(1,68);

   SetLevelValue(2,32);

   SetLevelStyle(STYLE_DOT,1,DimGray);

   

   return(0);

  }



int start()

  {

   double MA,RSI[];

   ArrayResize(RSI,Volatility_Band);

   int counted_bars=IndicatorCounted();

   int limit = Bars-counted_bars-1;

   for(int i=limit; i>=0; i--)

   {

      RSIBuf[i] = (iRSI(NULL,0,RSI_Period,RSI_Price,i)); 

      MA = 0;

      for(int x=i; x<i+Volatility_Band; x++) {

         RSI[x-i] = RSIBuf[x];

         MA += RSIBuf[x]/Volatility_Band;

      }

      UpZone[i] = (MA + (1.6185 * StDev(RSI,Volatility_Band)));

      DnZone[i] = (MA - (1.6185 * StDev(RSI,Volatility_Band)));  

      MdZone[i] = ((UpZone[i] + DnZone[i])/2);

      }

   for (i=limit-1;i>=0;i--)  

      {

       MaBuf[i] = (iMAOnArray(RSIBuf,0,RSI_Price_Line,0,RSI_Price_Type,i));

       MbBuf[i] = (iMAOnArray(RSIBuf,0,Trade_Signal_Line,0,Trade_Signal_Type,i));   

      } 

//----

   return(0);

  }

  

double StDev(double& Data[], int Per)

{return(MathSqrt(Variance(Data,Per)));

}

double Variance(double& Data[], int Per)

{double sum, ssum;

  for (int i=0; i<Per; i++)

  {sum += Data[i];

   ssum += MathPow(Data[i],2);

  }

  return((ssum*Per - sum*sum)/(Per*(Per-1)));

}

//+------------------------------------------------------------------+

DYNAMIC BREAKOUT SYSTEMS.docx

[bookmark: _GoBack]DYNAMIC BREAKOUT SYSTEMS Ever since the mid-eighties when Richard Dennis and Bill Eckhardt made a huge success with their turtle traders,' breakout systems have been on everybody's mind. The concept is easy enough: never mind trying to pick tops and bottoms-if if the market is going up, you buy, and if it is going down, you sell. In fact, instead of trying to sell the tops and buy the bottoms, do the opposite. That is, if the market  takes out the highest high for the last n days then buy, or if the market takes out the lowest low for the last n days then sell. With this basic system, you are in the market all the time, either with a long position anticipating an up-trend after a breakout on the long side, or a short position after a breakout on the short side. To this basic system several improvements or refinements can be made that could be used either separately or together with others. Among these is to treat the long and the short side separately, assigning them different lookback periods, or to have a separate lookback period handling the exits, which then results in the system's  also allowing for flat or neutral market situations in times of consolidation. Another refinement is to make the lookback period dynamic and sensitive to the historical volatility of the market. Yet another way is to keep the lookback period constant but add a volatility factor to the breakout level, which can be, for instance, the highest high for the last n days, times a volatility factor. One such volatility-sensitive system is the Dynamic Breakout System that has been featured in several articles for Futures magazine during 1996 and 1998. The logic behind the DBS system, as it is called for short, is that in times of relative  high volatility the number of "false" breakouts increases together with "false" trend reversals and deeper than expected corrections. Therefore, a volatility component increases the lookback period, to make it harder both to enter and exit a trade "too soon" The TradeStation code for the DBS system with a variable lookback back period of 20 to 60 days, follows: 





Inputs: MaxLB(60), MinLB(20): 


Vars: HistVol(0), YestHistVol(0), DeltaHistVol(0), EntryLB(0), ExitLB(0), 


YestEntryLB(0);


 YestHistVol = HistVol; 


HistVol = StdDev(C, 30); 


DeltaHistVol = (HistVol -YestHistVol) / HistVol: 


If CurrentBar = I Then 


EntryLB = 20; 


YestEntryLB = EntryLB;


EntryLB = YestEntryLB * (1 + DeltaHistVol):


 EntryLB = MaxList(EntryLB, MinLB); 


EntryLB = MinList(EntryLB, MaxLB);


 ExitLB = EntryLB * 0.5:


 Buy Tomorrow at Highest(High, EntryLB) Stop; 


Sell Tomorrow at Lowest(Low, EntryLB) Stop


 ExitLong Tomorrow at Lowest(Low, ExitLB) Stop;


 ExitShort Tomorrow at Highest(High, ExitLB) Stop; 





From the code (we will call this version la), you can see that the lookback period for the exits will always be half the lookback period for the entries at that particular point in time. The problem with this version, however, is that the variable able lookback period does not always work as intended. For instance, when traded on the CRB index over the period June 1986 to October 1999, and the Japanese yen over the period May 1972 to October 1999, the number of trades actually decreased somewhat for both markets, if the variable lookback period was substituted  for a stationary lookback period of 20 days. For the CRB index the number of trades decreased from 125 to 112 trades, and for the Japanese yen the same numbers were 191 with the variable lookback period and 181 without it. The good news is that, at least for the Japanese yen, the variable lookback period managed to produce a few more profitable trades-52% as compared to 46% with the 20-day day stationary lookback period. The same numbers for the CRB index were 26 and 28% respectively. Hence, the first part of this little study contradicts the logic behind the system, while the results from the second part can only be considered inconclusive. Another less desirable feature with this system is that with an increasing lookback period you run the risk of having the stops move away from the market instead of with 

